#include <SPI.h>
#include <MFRC522.h>

// ESP32C6的SPI引脚定义
#define SCK_PIN   6     // SPI时钟
#define MISO_PIN  5     // SPI主入从出
#define MOSI_PIN  7     // SPI主出从入
#define SS_PIN    10    // 片选引脚 (SDA)
#define RST_PIN   9     // 复位引脚

MFRC522 mfrc522(SS_PIN, RST_PIN);

void setup() {
  Serial.begin(115200);
  while (!Serial);
  
  // 明确初始化SPI，指定引脚
  SPI.begin(SCK_PIN, MISO_PIN, MOSI_PIN, SS_PIN);
  
  delay(100);
  
  // 初始化MFRC522
  mfrc522.PCD_Init();
  
  delay(100);
  
  Serial.println("=================================");
  Serial.println("    RFID-RC522 卡片读取系统");
  Serial.println("         XIUYUAN开发");
  Serial.println("=================================");
  Serial.println();
  
  // 检查连接
  byte version = mfrc522.PCD_ReadRegister(mfrc522.VersionReg);
  Serial.print("MFRC522 固件版本: 0x");
  Serial.println(version, HEX);
  
  if (version == 0x00 || version == 0xFF) {
    Serial.println("错误: 无法检测到MFRC522模块!");
    Serial.println("请检查接线:");
    Serial.println("  SDA  -> GPIO10");
    Serial.println("  SCK  -> GPIO6");
    Serial.println("  MOSI -> GPIO7");
    Serial.println("  MISO -> GPIO5");
    Serial.println("  RST  -> GPIO9");
    Serial.println("  GND  -> GND");
    Serial.println("  3.3V -> 3.3V");
    while(1);
  }
  
  Serial.println("MFRC522模块初始化成功!");
  Serial.println("请将IC卡靠近读卡器...");
  Serial.println();
}

void loop() {
  // 检查是否有新卡片
  if (!mfrc522.PICC_IsNewCardPresent()) {
    return;
  }
  
  // 选择一张卡片
  if (!mfrc522.PICC_ReadCardSerial()) {
    return;
  }
  
  // 显示分隔线
  Serial.println("=================================");
  Serial.println("检测到新卡片！");
  Serial.println("---------------------------------");
  
  // 显示卡片类型
  Serial.print("卡片类型: ");
  MFRC522::PICC_Type piccType = mfrc522.PICC_GetType(mfrc522.uid.sak);
  Serial.println(mfrc522.PICC_GetTypeName(piccType));
  
  // 显示卡片UID (十六进制格式)
  Serial.print("卡号 (HEX): ");
  for (byte i = 0; i < mfrc522.uid.size; i++) {
    Serial.print(mfrc522.uid.uidByte[i] < 0x10 ? " 0" : " ");
    Serial.print(mfrc522.uid.uidByte[i], HEX);
  }
  Serial.println();
  
  // 显示卡片UID (十进制格式)
  Serial.print("卡号 (DEC): ");
  for (byte i = 0; i < mfrc522.uid.size; i++) {
    Serial.print(mfrc522.uid.uidByte[i] < 10 ? " 00" : (mfrc522.uid.uidByte[i] < 100 ? " 0" : " "));
    Serial.print(mfrc522.uid.uidByte[i], DEC);
  }
  Serial.println();
  
  // 显示UID长度
  Serial.print("UID长度: ");
  Serial.print(mfrc522.uid.size);
  Serial.println(" 字节");
  
  // 将UID转换为单个数字（仅适用于4字节UID）
  if (mfrc522.uid.size == 4) {
    unsigned long cardID = 0;
    for (byte i = 0; i < 4; i++) {
      cardID = cardID * 256 + mfrc522.uid.uidByte[i];
    }
    Serial.print("卡号 (数字): ");
    Serial.println(cardID);
  }
  
  Serial.println("=================================");
  Serial.println();
  
  // 停止读卡
  mfrc522.PICC_HaltA();
  mfrc522.PCD_StopCrypto1();
  
  delay(1000);
}

